
keep if sample==1

foreach v of global all_DV { 
	display in red "`v'"
	qui reg `v' i.treatment##i.time $ControlsA if case_2!=1, vce(cluster agency)
		est store `v'_CA
	qui reg `v' i.treatment##i.time $ControlsB if case_2!=1, vce(cluster agency)
		est store `v'_CB
}

**#Panel A

esttab business_w5_CA comply_CA complete_CA open_request_CA extension_CA business_days_within_15_CA business_days_within_25_CA business_days_15_CA business_days_25_CA using "$Tables/Table_E6_a.tex", replace  label  b(%5.3f) se(%5.3f) star(* 0.10 ** 0.05 *** 0.01) mlabels("Business days" "Replied in time" "Complete answer" "Open request" "Extension was granted" "Replied within 15 bd" "Replied within 25 bd" "Replied in 15 bd" "Replied in 25 bd") refcat(1.treatment "\vspace{-6mm}", nolabel) stats(N r2,  labels(`"Observations"' `"R-squared"', prefix(\addlinespace[3pt])) fmt(%10.0fc %10.3fc 0 0) )  coeflabels($labels)   keep(1.treatment 1.time 1.treatment#1.time) order(1.treatment 1.time 1.treatment#1.time) nonumber nonote compress noomitted nobaselevels 


**#Panel B

#delimit ;
global Note_Table_E6 "
{\begin{minipage}{1.3\textwidth}
\small  
\vspace{0.2cm} 
\textit{Notes:} * p $math < $math 0.10, ** p $math < $math 0.05, *** p $math < $math 0.01. Standard errors are clustered by agency. The sample in the first column only considers requests that were closed by November 27, 2019, while the others take into account both pending and closed requests. Basic control variables in Panel A include: month in which the agency received the request, the total number of requests received by the agency before the intervention (Jan 1, 2017 until Sept. 5, 2018), average percentage of pending requests (not past the deadline - Mean of \% Pending Pre) before the intervention, and the average percentage of late requests prior to the intervention (Mean of \% Backlog Pre). In addition to those, all control variables in Panel B include the type of requester, serial requesters, topic of the request, total number of requests received by the agency on the date request \$ i \$ arrived, and ministry of the agency.
 \end{minipage}
 }
";
#delimit cr


esttab business_w5_CB comply_CB complete_CB open_request_CB extension_CB business_days_within_15_CB business_days_within_25_CB business_days_15_CB business_days_25_CB using "$Tables/Table_E6_b.tex", replace  label  b(%5.3f) se(%5.3f) star(* 0.10 ** 0.05 *** 0.01) mlabels("Business days" "Replied in time" "Complete answer" "Open request" "Extension was granted" "Replied within 15 bd" "Replied within 25 bd" "Replied in 15 bd" "Replied in 25 bd") refcat(1.treatment "\vspace{-6mm}", nolabel) stats(N r2,  labels(`"Observations"' `"R-squared"', prefix(\addlinespace[3pt])) fmt(%10.0fc %10.3fc 0 0) )  coeflabels($labels)  addnote($Note_Table_E6)  keep(1.treatment 1.time 1.treatment#1.time) order(1.treatment 1.time 1.treatment#1.time) nonumber nonote compress noomitted nobaselevels 
